

LASER™ PERSONAL COMPUTER 110

BASIC PROGRAMME



PROGRAMME IN BASIC

Die Informationen in diesem Heft können ohne vorhergehende Bekanntgabe verändert werden und stellen keine Verpflichtung für den Autor dar.

Alle Rechte vorbehalten.

Die Vervielfältigung und Übertragung auch einzelner Textabschnitte Bilder oder Zeichnungen ist — mit Ausnahme der Vervielfältigung zum persönlichen und eigenen Gebrauch gemäß § 53,54 URG — ohne schriftliche Zustimmung des Herausgebers nicht zulässig. Das gilt sowohl für Vervielfältigungen durch Fotokopie oder irgendein anderes Verfahren, als auch für die Übertragung auf Filme, Bänder, Platten, Arbeitstransparente oder andere Medien.

BASIC — PROGRAMMBEISPIELE

EINLEITUNG

Ihr Farbcomputer wurde geschaffen, um einem großen Personenkreis, also auch Anfängern, ohne vorhergehende Kenntnisse im Programmieren, Umgang mit Computern zu ermöglichen.

Dazu wurde die geeignete und am weitesten verbreitete Computersprache, BASIC, ausgewählt.

Am einfachsten lernt man das Programmieren, indem man Programme mit Hilfe des Computers schreibt.

Dieses Heft wurde geschrieben, um dem Bediener einige elementare Programmier-Techniken zu vermitteln, nachdem er sich mit den BASIC-Befehlen aus dem Referenz-Handbuch beschäftigt hat. Bitte beachten Sie, daß diese Programme relativ einfach sind und nicht die volle Leistungsfähigkeit dieses Farbcomputers demonstrieren. Der Autor empfiehlt dem Bediener deshalb, eigene Programme zu schreiben oder die gegebenen Programme zu verändern, um den Erfolg beim Programmieren mit dem Farbcomputer zu empfinden.

1) Summe & Durchschnitt	3
2) Permutation & Kombination	4
3) Größter gemeinsamer Teiler (GGT)	5
4) Kleinstes gemeinsames Vielfaches (KGV)	6
5) Primfaktorenzerlegung	7
6) Lösung quadratischer Gleichungen	8
7) Fläche von Dreiecken	9
8) Fläche von Vielecken	10
9) Winkel & Bogenmaß	11, 12
10) Fahrenheit und Celsius	13
11) Fuß und Meter	14
12) Pfund und Kilogramm	15
13) Gallonen und Liter	16
14) Wertverlust	17
15) Ordnen von Zahlen	18
16) Ordnen von Wörtern	19
17) Zahlenraten	20
18) Wörterraten	21
19) Zufallsgrafiken	22
20) Melodien	23
21) Zufallszahlen	24

1. SUMME UND DURCHSCHNITT

Dieses Programm berechnet die Summe und den Durchschnitt einer beliebigen Menge von Zahlen.

Können Sie den Programmablauf verfolgen?

```
10 REM SUMME UND DURCHSCHNITT
20 CLS
30 PRINT "SUMME UND DURCHSCHNITT" Zahl
40 INPUT "EINGABE: WIEVIEL ZAHLEN" A
50 FOR I = 1 TO A
60 PRINT "ZAHL"; I; "=";
70 INPUT B
80 C=C+B : NEXT
90 PRINT "SUMME ="; C
100 PRINT "DURCHSCHNITT ="; C/A
110 END
```

RUN

```
SUMME UND DURCHSCHNITT
EINGABE: WIEVIEL ZAHLEN? 5
ZAHL 1 =? 10
ZAHL 2 =? 20
ZAHL 3 =? 30
ZAHL 4 =? 40
ZAHL 5 =? 50
SUMME = 150
DURCHSCHNITT = 30
READY
```

2. PERMUTATION UND KOMBINATION

Permutation und Kombination sind zwei bekannte Bereiche der modernen Mathematik.

Durch dieses Programm können Sie die entsprechenden Antworten sehr schnell erhalten. Können Sie den Computer in Geschwindigkeit und Genauigkeit schlagen?

```
10 REM PERMUTATION UND
20 REM KOMBINATION
30 CLS
40 PRINT "PERMUTATION UND";
50 PRINT "KOMBINATION"
60 INPUT "GESAMTZahl DER ELEMENTE";A
70 PRINT "Zahl DER UNTERGEORDNETEN"
75 INPUT "ELEMENTE";B
80 C=1 : D=1
90 IF B>A THEN 30
100 FOR I = A-B+1 TO A
110 IF C*I>1E36 THEN 200
120 C=C*I : NEXT
130 FOR I = 2 TO B
140 D=D*I : NEXT
150 PRINT "PERMUTATION =" ; C
160 PRINT "KOMBINATION =" ; C/D
170 END
200 PRINT "UEBERLAUF" : GOTO 60
```

RUN

```
PERMUTATION UNDKOMBINATION
GESAMTZahl DER ELEMENTE? 5
Zahl DER UNTERGEORDNETEN
ELEMENTE? 4
PERMUTATION = 120
KOMBINATION = 5
READY
```

3. GRÖßTER GEMEINSAMER TEILER. (GGT)

Geben Sie zwei Zahlen ein, und der Computer liefert Ihnen den größten gemeinsamen Teiler.

```
10 REM BERECHNUNG DES GGT
20 CLS
30 PRINT "BERECHNUNG DES GGT"
40 INPUT "GEBEN SIE 2 ZAHLEN EIN ";A,B
50 IF A=0 OR B=0 THEN 100
60 IF A>B THEN A=A-B
70 IF A<B THEN B=B-A
80 IF A=>B THEN 60
90 PRINT "GGT =" ;A
100 END
```

RUN

```
BERECHNUNG DES GGT
GEBEN SIE 2 ZAHLEN EIN ? 20
?? 10
GGT = 10
READY
```

4. KLEINSTES GEMEINSAMES VIELFACHES. (KGV)

Ahnlich dem GGT wird mit diesem Program das Kleinste, gemeinsame Vielfache bestimmt.

```
10 REM BERECHNUNG DES KGV
20 CLS
30 PRINT "BERECHNUNG DES KGV"
40 INPUT "GEBEN SIE 2 ZAHLEN EIN ";A,B
50 IF A=0 OR B=0 THEN 110
60 IF A>B THEN C=A-1 ELSE C=B-1
70 C=C+1
80 IF INT(C/A)<>C/A THEN 70
90 IF INT(C/B)<>C/B THEN 70
100 PRINT "KGV =";C
110 END
```

RUN

```
BERECHNUNG DES KGV
GEBEN SIE 2 ZAHLEN EIN ? 11
?? 13
KGV = 143
READY
```

5. PRIMFAKTOREN-ZERLEGUNG

Dieses Program zerlegt eine beliebige Zahl in ihre Primfaktoren.

```
10 REM PRIMFAKTOREN-ZERLEGUNG
20 CLS
30 PRINT "PRIMFAKTOREN-ZERLEGUNG"
40 INPUT "GEBEN SIE EINE ZAHL EIN ";A
50 IF A=0 THEN 130
60 PRINT SGN(A); : A=ABS(A)
70 FOR I = 2 TO A : B=0
80 IF A/I<>INT(A/I) THEN 100
90 A=A/I : B=B+1 : GOTO 80
100 IF B=0 THEN 120
110 PRINT I;"^";B;
120 NEXT
130 END
```

RUN

```
PRIMFAKTOREN-ZERLEGUNG
GEBEN SIE EINE ZAHL EIN ? 240
1 2 ^ 4 3 ^ 1 5 ^ 1
READY
```

6. LÖSUNG QUADRATISCHER GLEICHUNGEN

Quadratische Gleichungen sind in der Form $ax^2 + bx + c = 0$ gegeben, wobei a, b und c die konstanten Koeffizienten und x die unbekannte Variable ist.

Dieses Programm liefert Ihnen die Lösung für x.

```
10 REM LOESUNG QUADRATISCHER
20 REM GLEICHUNGEN
30 CLS
40 PRINT "QUADRATISCHE GLEICHUNGEN"
50 PRINT "A * X ^ 2 + B * X + C = 0"
60 PRINT "GEBEN SIE DIE KOEFFIZIENTEN"
70 PRINT "A,B,C EIN"
80 INPUT A,B,C
90 D=B^2-4*A*C
100 IF D<0 THEN 160
110 D=SQR(D)
120 PRINT "DIE LOESUNGEN SIND :"
130 PRINT (-B-D)/(2*A);
140 PRINT (-B+D)/(2*A)
150 GOTO 170
160 PRINT "KEINE REELLE LOESUNG"
170 END
```

RUN

```
QUADRATISCHE GLEICHUNGEN
A * X ^ 2 + B * X + C = 0
GEBEN SIE DIE KOEFFIZIENTEN
A,B,C EIN
? 1
?? 1
?? -12
DIE LOESUNGEN SIND :
-4 3
READY
```

7. FLÄCHE VON DREIECKEN

Wenn alle drei Seiten eines Dreiecks bekannt sind, kann seine Fläche errechnet werden.

Können Sie selber nach diesem Programm ein Programm zur Berechnung des Umfangs eines Kreises bei gegebenem Radius entwerfen?

```
10 REM FLAECHE EINES DREIECKS
20 CLS
30 PRINT "FLAECHE EINES DREIECKS"
40 PRINT "GEBEN SIE DIE 3 SEITENLAENGEN"
50 INPUT "EIN";A,B,C
60 D=0.5*(A+B+C)
70 E=D*(D-A)*(D-B)*(D-C)
80 PRINT "DIE FLAECHE IST ";SQR(E)
90 END
```

RUN

```
FLAECHE EINES DREIECKS
GEBEN SIE DIE 3 SEITENLAENGEN
EIN? 6
?? 8
?? 10
DIE FLAECHE IST 24
READY
```

8. FLÄCHE VON VIELECKEN

Mit diesem Programm können sie die Fläche von regelmäßigen Vielecken berechnen.

Sie müßten nur die Anzahl der Ecken und die Länge einer Seite eingeben.

```
10 REM FLAECHE VON VIELECKEN
20 CLS : PI=3.1416
30 PRINT "FLAECHE VON REGELMAESSIGEN"
40 PRINT "VIELECKEN"
50 INPUT "ANZAHL DER SEITEN ";A
60 INPUT "SEITENLAENGE ";B
70 C=PI*(.5*A-1)/A
80 D=A*B*B*TAN(C)/4
90 PRINT "FLAECHE =" ;D
100 END
```

RUN

```
FLAECHE VON REGELMAESSIGEN
VIELECKEN
ANZAHL DER SEITEN ? 5
SEITENLAENGE ? 4
FLAECHE = 27.5278
READY
```

9. WINKEL UND BOGENMAß.

Dieses Programm rechnet Werte im Winkelmaß in das Bogenmaß, und umgekehrt, um.

```
10 REM WINKELMASS UND BOGENMASS
20 CLS
30 PRINT "GESUCHT: 1=BOGENMASS"
35 INPUT "          2=WINKELMASS"; S
40 IF S=1 THEN 140
50 INPUT "BOGENMASS"; B
60 C=B*180/3.1416
70 IF C>360 THEN C=C-360 : GOTO 70
80 PRINT INT(C); "GRAD"
90 D=(C-INT(C))*60
100 PRINT INT(D); "MINUTEN"
110 E=(D-INT(D))*60
120 PRINT INT(E); "SEKUNDEN"
130 END
140 INPUT "GRAD"; A
150 INPUT "MINUTEN"; B
160 INPUT "SEKUNDEN"; C
170 PRINT
180 D=A+B/60+C/3600
190 IF D>360 THEN D=D-360 : GOTO 190
200 D=D*3.1416/180
210 PRINT D; "RAD"
220 END
```

RUN

```
GESUCHT: 1=BOGENMASS
          2=WINKELMASS? 1
GRAD? 1
MINUTEN? 1
SEKUNDEN? 1
```

```
.0177491 RAD
READY
```

RUN

GESUCHT: 1=BOGENMASS
2=WINKELMASS? 2

BOGENMASS? 1

57 GRAD

17 MINUTEN

44 SEKUNDEN

READY

10. FAHRENHEIT UND CELSIUS

Ähnlich wie bei "Winkelmaß und Bogenmaß" rechnet dieses Programm Temperaturgrade um.

```
10 REM GRAD FAHRENHEIT UND
20 REM CELSIUS
30 CLS
40 PRINT "GESUCHT: 1=GRAD F"
50 INPUT "    ODER  2=GRAD C";A
60 IF A=2 THEN 110
70 INPUT "GRAD C =";B
80 PRINT B;"GRAD C =";
90 PRINT B*9/5+32;"GRAD F"
100 END
110 INPUT "GRAD F =";B
120 PRINT B;"GRAD F =";
130 PRINT (B-32)*5/9;"GRAD C"
140 END
```

RUN

```
GESUCHT: 1=GRAD F
        ODER  2=GRAD C? 1
GRAD C =? 0
  0 GRAD C = 32 GRAD F
READY
```

RUN

```
GESUCHT: 1=GRAD F
        ODER  2=GRAD C? 2
GRAD F =? 32
  32 GRAD F = 0 GRAD C
READY
```

11. FUß UND METER

Ähnlich wie in "Winkel- und Bogenma " werden hier Fu in Meter, und umgekehrt, umgerechnet.

```
10 REM FUSS UND METER
20 CLS
30 PRINT "GESUCHT: 1=FUSS"
40 INPUT "      ODER  2=METER";A
50 IF A=1 THEN 100
60 INPUT "FUSS";B
70 PRINT B;"FUSS =";
80 PRINT .3048*B;"METER"
90 END
100 INPUT "METER";B
110 PRINT B;"METER =";
120 PRINT B/.3048;"FUSS"
130 END
```

RUN

```
GESUCHT: 1=FUSS
      ODER  2=METER? 1
METER? 1
 1 METER = 3.28084 FUSS
READY
```

RUN

```
GESUCHT: 1=FUSS
      ODER  2=METER? 2
FUSS? 1
 1 FUSS = .3048 METER
READY
```

12 PFUND & KILOGRAMM

Ähnlich wie bei "Winkel- und Bogenmaß" werden hier Pfund in Kilogramm, und umgekehrt, umgerechnet.

```
10 REM PFUND & KILOGRAMM
20 CLS
30 PRINT "BERECHNUNG DER PFUND (1)"
35 INPUT " ODER DER KILOGRAMM (2)";A
40 IF A=1 THEN 90
50 INPUT "PFUND";B
60 PRINT B;"PFUND =" ;
70 PRINT .4536*B;"KILOGRAMM"
80 END
90 INPUT "KILOGRAMM";B
100 PRINT B;"KILOGRAMM =" ;
110 PRINT B/.4536;"PFUND"
120 END
```

RUN

```
BERECHNUNG DER PFUND (1)
 ODER DER KILOGRAMM (2)? 1
KILOGRAMM? 1
 1 KILOGRAMM = 2.20459 PFUND
READY
```

RUN

```
BERECHNUNG DER PFUND (1)
 ODER DER KILOGRAMM (2)? 2
PFUND? 1
 1 PFUND = .4536 KILOGRAMM
READY
```

13. GALLONEN & LITER

Ähnlich wie bei "Winkel- und Bogenma " werden hier Gallonen in Liter, und umgekehrt, umgerechnet.

```
10 REM GALLONEN & LITER
20 CLS
30 PRINT "BERECHNUNG DER GALLONEN (1)"
35 INPUT "          ODER DER LITER (2)";A
40 IF A=1 THEN 90
50 INPUT "GALLONEN";B
60 PRINT B;"GALLONEN =" ;
70 PRINT 3.785*B;"LITER"
80 END
90 INPUT "LITER";B
100 PRINT B;"LITER =" ;
110 PRINT B/3.785;"GALLONEN"
120 END
```

RUN

```
BERECHNUNG DER GALLONEN (1)
          ODER DER LITER (2)? 1
LITER? 1
 1 LITER = .264201 GALLONEN
READY
```

RUN

```
BERECHNUNG DER GALLONEN (1)
          ODER DER LITER (2)? 2
GALLONEN? 1
 1 GALLONEN = 3.785 LITER
READY
```

14. WERTVERLUST

Der Wert einer jeden Ware sinkt mit der Zeit.

Dieses Programm berechnet den absoluten Wertverlust (die Differenz zum Neupreis), unter Berücksichtigung des Wertverlustes in Prozent über einen bestimmten Zeitraum.

```
10 REM WERTVERLUST
20 CLS
30 INPUT "NEUPREIS";A
40 INPUT "WERTVERLUST (%) PRO JAHR";B
50 INPUT "ANZAHL DER JAHRE";C
60 PRINT "WERTVERLUST =" ;
70 B=B/100
80 D=A*B*(1-B)^(C-1)
90 D=INT(D*10+.5)/10
100 PRINT D : END
```

RUN

```
NEUPREIS? 1000
WERTVERLUST (%) PRO JAHR? 10
ANZAHL DER JAHRE? 5
WERTVERLUST = 65.6
READY
```

15. ORDNEN VON ZAHLEN

Geben Sie eine Menge von Zahlen zwischen 2 und 20 ein, so ordnet dieses Programm die Zahlen in ansteigender Reihenfolge. Können Sie das Programm so ändern, daß die Zahlen in absinkender Reihenfolge sortiert werden?

```
10 REM ORDNEN VON ZAHLEN IN
20 REM ANSTEIGENDER REIHENFOLGE
30 CLS
40 PRINT "ORDNEN VON ZAHLEN (2-20)"
50 INPUT "WIEVIELE ZAHLEN";A
60 DIM A(19)
70 FOR I= 1 TO A
80 PRINT "ZAHL";I; : INPUT A(I-1)
90 NEXT
100 FOR J = 0 TO A-2
110 FOR I = 0 TO A-2
120 IF A(I)<A(I+1) THEN 140
130 B=A(I) : A(I)=A(I+1) : A(I+1)=B
140 NEXT : NEXT
150 FOR I = 0 TO A-1
160 PRINT A(I);
170 NEXT
180 END
```

RUN

ORDNEN VON ZAHLEN (2-20)

WIEVIELE ZAHLEN? 6

ZAHL 1 ? 6

ZAHL 2 ? 5

ZAHL 3 ? 4

ZAHL 4 ? 3

ZAHL 5 ? 2

ZAHL 6 ? 1

1 2 3 4 5 6

READY

16. ORDNEN VON WÖRTERN

Das Programm ordnet eine Anzahl von Wörtern (von 2-10) in alphabetischer Reihenfolge.

```
10 REM ORDNEN VON WÖRTERN IN
20 REM ALPHABETISCHER
25 REM REIHENFOLGE
30 CLS
40 PRINT "ORDNEN VON WÖRTERN (2-10)"
50 INPUT "WIEVIELE WÖRTER";A
60 DIM A$(9)
70 FOR I= 1 TO A
80 PRINT "WORT";I; : INPUT A$(I-1)
90 NEXT
100 FOR J= 0 TO A-2
110 FOR I= 0 TO A-2
120 IF A$(I)<A$(I+1) THEN 140
130 B$=A$(I) : A$(I)=A$(I+1) : A$(I+1)=B$
140 NEXT : NEXT
150 FOR I = 0 TO A-1
160 PRINT A$(I); " ";
170 NEXT
180 END
```

RUN

```
ORDNEN VON WÖRTERN (2-10)
WIEVIELE WÖRTER? 6
WORT 1 ? WORT
WORT 2 ? VASE
WORT 3 ? KOHL
WORT 4 ? LAUF
WORT 5 ? DACH
WORT 6 ? HAUS
DACH HAUS KOHL LAUF VASE WORT
READY
```

17. ZAHLENRATEN

Der Computer bestimmt eine Zufallszahl zwischen 1 und 1000. Sie müssen diese Zahl erraten.

Wieviele Versuche haben Sie gebraucht?

```
10 REM ZAHLENRATEN
20 CLS : C=1
30 A=RND(1000)
40 PRINT "ERRATEN SIE EINE ZAHL"
50 INPUT "(1-1000)";B
60 IF B>A THEN PRINT "KLEINER"
70 IF B<A THEN PRINT "GROESSER"
80 IF B=A THEN 100
90 C=C+1 : GOTO 40
100 PRINT "SIE HABEN DIE ZAHL ERRATEN"
110 PRINT "SIE HABEN";C;
120 PRINT "VERSUCHE BENOETIGT"
130 END
```

RUN

```
ERRATEN SIE EINE ZAHL
(1-1000)? 500
KLEINER
ERRATEN SIE EINE ZAHL
(1-1000)? 250
GROESSER
ERRATEN SIE EINE ZAHL
(1-1000)? 300
SIE HABEN DIE ZAHL ERRATEN
SIE HABEN 3 VERSUCHE BENOETIGT
READY
```

18. WÖRTERRATEN

Diesesmal haben Sie ein Wort aus 4 Buchstaben zu erraten. Die Methode des Spielens ist ähnlich der des Zahlensvatens.

```
10 REM WÖRTERRATEN
20 CLS
30 C$="WORTVASEKOHLLAUFDACHHAUSMUNDZIELTOPFOFEN"
40 I=(RND(10)-1)*4+1
50 A$=MID$(C$,I,4) : S=1
60 PRINT "WÖRTERRATEN"
70 INPUT "(4 BUCHSTABEN)";B$
80 FOR J = 1 TO 4
90 IF MID$(A$,1,J)=MID$(B$,1,J) THEN NEXT
100 PRINT "SIE HABEN";J-1;
110 PRINT "BUCHSTABEN RICHTIG GERATEN"
120 IF J<>5 THEN S=S+1 : GOTO 60
130 PRINT "SIE HABEN";S;
140 PRINT "VERSUCHE BENOETIGT"
150 END
```

RUN

WÖRTERRATEN

(4 BUCHSTABEN)? W

SIE HABEN 0 BUCHSTABEN RICHTIG
GERATEN

WÖRTERRATEN

(4 BUCHSTABEN)? V

SIE HABEN 1 BUCHSTABEN RICHTIG
GERATEN

WÖRTERRATEN

(4 BUCHSTABEN)? VASE

SIE HABEN 4 BUCHSTABEN RICHTIG
GERATEN

SIE HABEN 3 VERSUCHE BENOETIGT

READY

19. ZUFALLSGRAFIKEN

Zufallsgrafiken werden erzeugt, indem die zuvor festgelegten grafischen Zeichen beliebig ausgedruckt werden.

```
10 REM GRAFIK
20 CLS
30 COLOR RND(8)
40 PRINT@ RND(512)-1, "■■";
50 GOTO 30
```

20. MELODIEN

Sie können Ihre eigenen Lieder schreiben und spielen. Sie müssen hierzu nur den Frequenzcode und die Dauer der einzelnen Töne wählen. Die Anzahl der Noten, die Sie spielen können, hängt von der Speicherplatzkapazität Ihres Computers ab.

```
10 REM LIEDER
20 CLS
30 PRINT "GEBEN SIE DIE ANZAHL DER"
35 INPUT "NOTEN EIN";N
40 PRINT "GEBEN SIE IHRE NOTEN EIN"
50 DIM A%(2*N-1)
60 FOR I= 0 TO N-1
70 INPUT "FREQUENZCODE";A%(I*2)
80 INPUT "LAENGENCODE";A%(I*2+1)
90 NEXT
100 FOR I= 0 TO N-1
110 SOUND A%(I*2),A%(I*2+1)
120 NEXT
```

RUN

```
GEBEN SIE DIE ANZAHL DER
NOTEN EIN? 8
GEBEN SIE IHRE NOTEN EIN
FREQUENZCODE? 26
LAENGENCODE? 3
FREQUENZCODE? 30
LAENGENCODE? 3
FREQUENZCODE? 28
LAENGENCODE? 3
FREQUENZCODE? 21
LAENGENCODE? 5
FREQUENZCODE? 26
LAENGENCODE? 3
FREQUENZCODE? 28
LAENGENCODE? 3
FREQUENZCODE? 30
LAENGENCODE? 3
FREQUENZCODE? 26
LAENGENCODE? 7
READY
```

21. ZUFALLSZAHLEN

Der Computer bestimmt 6 Zufallszahler und eine Zusatzzahl.

```
10 REM ZUFALLSZAHLEN
20 CLS
30 FOR I = 1 TO 7
40 A(I)=RND(36)
50 IF I = 1 THEN 90
60 FOR J = 1 TO I-1
70 IF A(I)=A(J) THEN 40
80 NEXT
90 NEXT
100 PRINT "DIE ZAHLEN LAUTEN :"
110 FOR I = 1 TO 5
120 FOR J = 1 TO 5
130 IF A(J)<A(J+1) THEN 150
140 B=A(J) : A(J)=A(J+1) : A(J+1)=B
150 NEXT : NEXT
160 FOR J = 1 TO 6
170 PRINT A(J);
180 NEXT
190 PRINT
200 PRINT "DIE ZUSATZZAHL LAUTET :"
210 PRINT A(7)
220 END
```

RUN

DIE ZAHLEN LAUTEN :

5 10 15 20 26 35

DIE ZUSATZZAHL LAUTET :

17

READY

© 1983 VTL. MADE IN HONG KONG

91-0123-05